

Chapter 16

Performing Quick LDM Evaluations

Contents

Performing Quick LDM Evaluations..... 16-3

 Command Mnemonics..... 16-3

 Evaluating Optical Values 16-3

 What You Need to Know About Data Evaluation (EVA Command)..... 16-4

Performing Quick LDM Evaluations


The LDM quick evaluation feature allows you to evaluate single system quantities or any expression, using Macro-PLUS.

Command Mnemonics (alphabetical)

EVA

Evaluating Optical Values

To perform a quick evaluation of optical values in your lens system using a Macro-PLUS expression, choose the **Tools > Evaluate Expression** menu. This displays the **Evaluate Expression** dialog box. Define the controls in this dialog box as described in the table below.



Command Syntax	
Screen Control	Explanation
EVA macro_expression	
Tools > Evaluate Expression menu	Evaluates and lists the value of the Macro-PLUS expression. See “Expressions” on page 25A-95 for expression formation. In the GUI, enter the Macro-PLUS expression in the Macro Expression field. Both in the GUI and on the command line, remember to enclose database items in parentheses. Examples: In the GUI: (EFY S1..5) On the command line: EVA (EFY S1..5) – Calculates EFL for part of the system (surfaces 1 through 5) In the GUI: 2*(NA)/(WL WL) On the command line: EVA 2*(NA)/(WL WL) – Calculates diffraction cutoff frequency for shortest (last) wavelength (default MFR in MTF option)

In the GUI, once you enter an expression in the **Macro Expression** field, click the **Evaluate** button. The value of the expression is then displayed in the **Result** field. An error message is displayed if your macro expression syntax is incorrect.

What You Need to Know About Data Evaluation (EVA Command)

The **Evaluate Expression** dialog box (**Tools** menu, or **EVA** command) allows display of virtually any part of the lens database, and allows calculations to be made with these database items or with user-defined variables. Refer to “Database Items” on page 25B-3 for a description of the database items that can be accessed by **EVA**. Refer to “Expressions” on page 25A-95 for syntax rules on expressions that can be evaluated by **EVA** and for a description of the pre-defined functions that can be used in expressions.

The general form of the **EVA** command is **EVA** followed by a Macro-PLUS expression. There are no limits to what can be in a macro expression, as long as it follows the standard rules. If the expression to be evaluated is longer than a single line, an ampersand (&) can be placed at the end of a line to indicate continuation on the next line (multiple lines are allowed).

In the GUI, the output of the EVA command is displayed in the **Result** field in the **Evaluate Expression** dialog box. On the command line, the output of an **EVA** command is a repeat of the requested expression followed by the result. This result may be a number or a text string. If the results are logical values, they are given as 1 for TRUE and 0 for FALSE (for example, **EVA (NUM F)>3** will result in 1 if there are four or five fields specified, or 0 if there are three fields or less specified).

Note that all CODE V database items must be enclosed in parentheses (even if they are used inside other parentheses, such as a function argument) and note that all trigonometric calculations assume that angles are in radians. As in the rest of CODE V, expressions ignore case (upper or lower), except in text strings.

Examples:

```
EVA (EFL)           ! Evaluate the effective focal length of the lens
EVA (EFL Z2)        ! Evaluate the focal length for the second zoom
                    ! position
EVA SIN((ADE S2)/57.295)           ! Angle must be in radians
EVA (HMY S4)*(CUY S7)/LOGF((FNO)) ! Arbitrary calculation
```

The **EVA** command can be used as a general purpose calculator (the expression does not have to include database items):

```
EVA SQRTF(28.345)
EVA 10*4/((16*4+3)-SINF(.27))
```

The **EVA** command can access ray trace data and perform calculations on them. If the R and F qualifiers are used, then the corresponding reference ray datum is used. If the R and F qualifiers are not used, then the datum is for the last ray traced. The reference wavelength is assumed unless specified by the W qualifier, and the first zoom position is assumed unless specified by the Z qualifier.

```
EVA (Y F1 R2 SI)           ! Evaluate the Y value at the image surface (SI) of
                          ! the upper marginal ray (R2)for field 1 (F1)
EVA (Y R1 FL SI W3 Z3)-(Y R1 FL SI W2 Z3)
                          ! Evaluate real lateral color (Y difference
                          ! for chief ray, R1) between the 2nd and 3rd
                          ! wavelength, at the last field (FL)
                          ! for the third zoom position (Z3)
EVA (M S3)/(N S3)          ! Evaluate the ray Y direction tangent (M/N) after
                          ! surface 3 for the last ray traced
```

The output of the EVA command is given to 12 significant digits (if needed). This can provide extra precision for database items which are listed to less precision in standard CODE V output.

```
SUR S2                                ! Lists the radius to 5 digits (after the decimal point)
EVA (RDY S2)                          ! Lists the radius to 12 digits (if needed)
```

If any user-defined variables have been declared by you, they can be used in **EVA** expressions. Refer to “User-Defined Elements” on page 25A-7 for a description of Macro-PLUS variables and their use.

```
^RAD == 45/ATANF(1)                  ! Define variable ^rad to be 1 radian (57.295...)
EVA SIN(30/^RAD)                     ! Converts degrees to radians for trigonometric
                                     ! calculation
```

